
;г====================================¬

;¦          НАЧАЛО ПРОГРАММЫ          ¦

;¦ (ЗАГРУЗКА ФАЙЛОВ И ИНИЦИАЛИЗАЦИЯ)  ¦

;L====================================-

FLAG  equ 37900

      ent 26000

      org 26000

      xor  a          ;ОЧИСТИЛИ
      ld   (CLSATR),a ;ЭКРАН
      call CLS        ;

      ld   hl,44000   ;

      ld   (LDADR),hl ;ЗАГРУЗИЛИ
      ld   hl,FILEZ1  ;ЗАСТАВКУ #1

      call LOAD       ;

      ld   hl,51000   ;

      ld   (LDADR),hl ;ЗАГРУЗИЛИ
      ld   hl,FILEZ2  ;ЗАСТАВКУ #2

      call LOAD       ;

      ld   hl,58000   ;

      ld   (LDADR),hl ;ЗАГРУЗИЛИ
      ld   hl,FILEZ3  ;ЗАСТАВКУ #3

      call LOAD       ;

      ld   b,4

      call RAM

      ld   hl,#C000   ;

      ld   (LDADR),hl ;ЗАГРУЗИЛИ
      ld   hl,FILEZ4  ;МЕНЮ
      call LOAD       ;

      ld   hl,57000   ;

      ld   (LDADR),hl ;ЗАГРУЗИЛИ
      ld   hl,FILEOS  ;ОСН ЭКРАН
      call LOAD       ;

      ld   b,3

      call RAM

      ld   hl,#C000   ;

      ld   (LDADR),hl ;ЗАГРУЗИЛИ
      ld   hl,FILEZ5  ;СИСТЕМНЫЕ
      call LOAD       ;СПРАЙТЫ
      ld   b,0

      call RAM

      ld   hl,44000

      ld   (BUFF),hl 

      call FI :call PAUSE :call FO

      ld   hl,51000

      ld   (BUFF),hl 

      call FI :call PAUSE :call FO

      ld   hl,58000

      ld   (BUFF),hl 

      call FI :call PAUSE :call FO

      ld  hl,0        ;ИНИЦ ПЕРЕМЕННЫЕ

      ld  (WINBF1),hl ;ДЛЯ ВЫРЕЗКИ ОКОН

      ld  (WINBF2),hl ;КРОМЕ 0-ОГО,ОНО

      ld  (WINBF3),hl ;ВСЕГДА #C000

      xor a           ;

      ld  hl,32768    ;ОЧИСТИЛИ ВСЮ

      ld  (hl),a      ;ПАМЯТЬ ПОД

      ld  de,32769    ;ДАННЫЕ
      ld  bc,32766    ;

      ldir            ;

      ld   hl,47000   ;ЗАДАЛИ АДРЕС
      ld   (NABOR),hl ;ШРИФТА И ЗАГУ-

      ld   (LDADR),hl ;ЗИЛИ ЕГО В ЭТОТ

      ld   hl,FILE01  ;АДРЕС
      call LOAD

      ld   hl,#C000   ;УСТАНОВИЛИ УКАЗА-

      ld   (POZGL),hl ;ТЕЛЬ В БЛОКЕ

      ld   (LDADR),hl ;СЦЕН НА НАЧАЛО И

      ld   hl,FILE03  ;ЗАГРУЗИЛИ
      call LOAD       ;БЛОК СЦЕН #00

;г====================================¬

;¦      ОСНОВНОЙ БЛОК ПРОГРАММЫ       ¦

;L====================================-

;------ ПЕРЕБРОСКА ТЕКУЩЕЙ СЦЕНЫ ------

;------  В БУФЕР ОБРАБОТКИ СЦЕН  ------

GLV00   ld hl,(POZGL)

        ld de,#8000

GLV01   ld a,(hl) 

        cp #FE

        jr z,GLV02

        ld (de),a 

        inc hl

        inc de 

        jr  GLV01

GLV02   ld  (de),a 

        ld  de,#8000

        ld  (POZIC),de 

;г====================================¬

;¦    БЛОК ОБРАБОТКИ ТЕКУЩЕЙ СЦЕНЫ    ¦

;¦   (РАСПОЗНАНИЯ УПРАВЛЯЮЩИХ КОДОВ)  ¦

;L====================================-

OSN00 ;WWN - Work WiNdow ++++++++++++++

      ld  hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN01

      inc hl  : ld  a,(hl) 

      cp  "W" : jr nz,OSN01

      inc hl  : ld  a,(hl) 

      cp  "N" : jr nz,OSN01

              call COD00

              inc hl     ;НА СЛЕДУЮЩ КОД

              ld  (POZIC),hl 

              jp  OSN00

OSN01 ;WC1 - Window Clear1 +++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN02

      inc hl  : ld  a,(hl) 

      cp  "C" : jr nz,OSN02

      inc hl  : ld  a,(hl) 

      cp  "1" : jr nz,OSN02

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD01     ;НЕТ
              jp  OSN00

OSN02 ;WC2 - Window Clear2 +++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN03

      inc hl  : ld  a,(hl) 

      cp  "C" : jr nz,OSN03

      inc hl  : ld  a,(hl) 

      cp  "2" : jr nz,OSN03

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD02     ;НЕТ
              jp  OSN00

OSN03 ;WCL - Window Color  +++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN04

      inc hl  : ld  a,(hl) 

      cp  "C" : jr nz,OSN04

      inc hl  : ld  a,(hl) 

      cp  "L" : jr nz,OSN04

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD03     ;НЕТ
              jp  OSN00

OSN04 ;WRM - Window Ramka  +++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN05

      inc hl  : ld  a,(hl) 

      cp  "R" : jr nz,OSN05

      inc hl  : ld  a,(hl) 

      cp  "M" : jr nz,OSN05

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD04     ;НЕТ
              jp  OSN00

OSN05 ;WCT - Window CuT  +++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN06

      inc hl  : ld  a,(hl) 

      cp  "C" : jr nz,OSN06

      inc hl  : ld  a,(hl) 

      cp  "T" : jr nz,OSN06

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD05     ;НЕТ
              jp  OSN00

OSN06 ;WST - Window set  +++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN07

      inc hl  : ld  a,(hl) 

      cp  "S" : jr nz,OSN07

      inc hl  : ld  a,(hl) 

      cp  "T" : jr nz,OSN07

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD06     ;НЕТ
              jp  OSN00

OSN07 ;WPT - Window PuT  +++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN08

      inc hl  : ld  a,(hl) 

      cp  "P" : jr nz,OSN08

      inc hl  : ld  a,(hl) 

      cp  "T" : jr nz,OSN08

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD07     ;НЕТ
              jp  OSN00

OSN08 ;WN1 - Window #1  +++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN09

      inc hl  : ld  a,(hl) 

      cp  "N" : jr nz,OSN09

      inc hl  : ld  a,(hl) 

      cp  "1" : jr nz,OSN09

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD08     ;НЕТ
              jp  OSN00

OSN09 ;W2O - Window 2 Open +++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN10

      inc hl  : ld  a,(hl) 

      cp  "2" : jr nz,OSN10

      inc hl  : ld  a,(hl) 

      cp  "O" : jr nz,OSN10

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD09     ;НЕТ
              jp  OSN00

OSN10 ; ";"- КОММЕНТАРИЙ  ++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  ";" : jr nz,OSN11

OSN101        inc hl 

              ld  a,(hl)

              cp  13

              jr  nz,OSN101

              inc hl 

              ld  (POZIC),hl 

              jp   OSN00

OSN11 ; ПЕРЕВОД СТРОКИ  ++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp 13   : jr nz,OSN12

              inc  hl

              ld   (POZIC),hl 

              jp   OSN00

OSN12 ;BNK - BaNK  +++++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "B" : jr nz,OSN13

      inc hl  : ld  a,(hl) 

      cp  "N" : jr nz,OSN13

      inc hl  : ld  a,(hl) 

      cp  "K" : jr nz,OSN13

              inc  hl ;ПРОПУСТ ПРОБЕЛ
              inc  hl 

              ld   a,(hl)

              sub  48

              ld   b,a 

              inc  hl ;ПРОП ПЕРЕВОД СТР
              inc  hl 

              ld   (POZIC),hl 

              call RAM

              jp   OSN00

OSN13 ;PAU - PAUSE  ++++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "P" : jr nz,OSN14

      inc hl  : ld  a,(hl) 

      cp  "A" : jr nz,OSN14

      inc hl  : ld  a,(hl) 

      cp  "U" : jr nz,OSN14

              inc hl ; ПРОП ПЕРЕВ СТР
              inc hl 

              ld (POZIC),hl 

              call PAUSE

              jp OSN00

OSN14 ;TXT - TeXT  +++++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "T" : jr nz,OSN15

      inc hl  : ld  a,(hl) 

      cp  "X" : jr nz,OSN15

      inc hl  : ld  a,(hl) 

      cp  "T" : jr nz,OSN15

              inc hl       ;ППСТР
              inc hl 

              ld (POZIC),hl 

              call COD14

              jp OSN00

OSN15 ;CLS - Clear Screen  +++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "C" : jr nz,OSN16

      inc hl  : ld  a,(hl) 

      cp  "L" : jr nz,OSN16

      inc hl  : ld  a,(hl) 

      cp  "S" : jr nz,OSN16

              inc hl  ;ПРОП ПРОБЕЛ
              inc hl

              ld  a,3

              ld  (RAZR),a 

              call CONV

              ld  (CLSATR),a 

              inc hl         ;ППС
              ld (POZIC),hl 

              call CLS

              jp OSN00

OSN16 ;FDI - FaDe in  ++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "F" : jr nz,OSN17

      inc hl  : ld  a,(hl) 

      cp  "D" : jr nz,OSN17

      inc hl  : ld  a,(hl) 

      cp  "I" : jr nz,OSN17

              inc hl ;ПЕРЕШЛИ НА ПРОБЕЛ

              inc hl ;ПЕРЕШЛИ НА ЧИСЛО

              ld  a,5

              ld  (RAZR),a 

              call CONV

              inc hl

              ld  (POZIC),hl

              ld  hl,(CONVR)

              ld  (BUFF),hl 

              call FI

              jp   OSN00

OSN17 ;FDO - FaDe out  +++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "F" : jr nz,OSN18

      inc hl  : ld  a,(hl) 

      cp  "D" : jr nz,OSN18

      inc hl  : ld  a,(hl) 

      cp  "O" : jr nz,OSN18

              inc  hl ;ППС
              inc  hl 

              ld   (POZIC),hl 

              call FO

              jp   OSN00

OSN18 ;CSP - Cut SPait  ++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "C" : jr nz,OSN19

      inc hl  : ld  a,(hl) 

      cp  "S" : jr nz,OSN19

      inc hl  : ld  a,(hl) 

      cp  "P" : jr nz,OSN19

              call COD18

              inc hl     ;НА СЛЕДУЮЩ КОД

              ld  (POZIC),hl 

              jp  OSN00

OSN19 ;PSP - Put SPait  ++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "P" : jr nz,OSN20

      inc hl  : ld  a,(hl) 

      cp  "S" : jr nz,OSN20

      inc hl  : ld  a,(hl) 

      cp  "P" : jr nz,OSN20

              call COD19

              inc hl     ;НА СЛЕДУЮЩ КОД

              ld  (POZIC),hl 

              jp  OSN00

OSN20 ;SP1 - SPait #1 ++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "S" : jr nz,OSN21

      inc hl  : ld  a,(hl) 

      cp  "P" : jr nz,OSN21

      inc hl  : ld  a,(hl) 

      cp  "1" : jr nz,OSN21

              call COD20

              inc hl     ;НА СЛЕДУЮЩ КОД

              ld  (POZIC),hl 

              jp  OSN00

OSN21 ;SP2 - SPait #2 ++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "S" : jr nz,OSN22

      inc hl  : ld  a,(hl) 

      cp  "P" : jr nz,OSN22

      inc hl  : ld  a,(hl) 

      cp  "2" : jr nz,OSN22

              call COD21

              inc hl     ;НА СЛЕДУЮЩ КОД

              ld  (POZIC),hl 

              jp  OSN00

OSN22 ;LDF - LOAD FILE +++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "L" : jr nz,OSN23

      inc hl  : ld  a,(hl) 

      cp  "D" : jr nz,OSN23

      inc hl  : ld  a,(hl) 

      cp  "F" : jr nz,OSN23

              call COD22

              inc hl     ;НА СЛЕДУЮЩ КОД

              ld  (POZIC),hl 

              jp  OSN00

OSN23 ;JMP - JUMP ++++++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "J" : jr nz,OSN24

      inc hl  : ld  a,(hl) 

      cp  "M" : jr nz,OSN24

      inc hl  : ld  a,(hl) 

      cp  "P" : jr nz,OSN24

            call COD23

            jp   GLV00  ;!!!!!-GLV!!!

OSN24 ;VIB - VIBOR +++++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "V" : jr nz,OSN25

      inc hl  : ld  a,(hl) 

      cp  "I" : jr nz,OSN25

      inc hl  : ld  a,(hl) 

      cp  "B" : jr nz,OSN25

            inc  hl ;ПЕРЕШЛИ НА ПРОБЕЛ

            inc  hl ;НА КОЛИЧЕСТВО ВАРИАНТО ВЫБОРА

            call COD24

            jp   GLV00  ;!!!!!-GLV!!!

OSN25 ;W2C - Window 2 Close ++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN26

      inc hl  : ld  a,(hl) 

      cp  "2" : jr nz,OSN26

      inc hl  : ld  a,(hl) 

      cp  "C" : jr nz,OSN26

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD25     ;НЕТ
              jp  OSN00

OSN26 ;W3O - Window 3 Open +++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN27

      inc hl  : ld  a,(hl) 

      cp  "3" : jr nz,OSN27

      inc hl  : ld  a,(hl) 

      cp  "O" : jr nz,OSN27

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD26     ;НЕТ
              jp  OSN00

OSN27 ;W2C - Window 3 Close ++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN28

      inc hl  : ld  a,(hl) 

      cp  "3" : jr nz,OSN28

      inc hl  : ld  a,(hl) 

      cp  "C" : jr nz,OSN28

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD27     ;НЕТ
              jp  OSN00

OSN28 ;FON - FlagON ++++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "F" : jr nz,OSN29

      inc hl  : ld  a,(hl) 

      cp  "O" : jr nz,OSN29

      inc hl  : ld  a,(hl) 

      cp  "N" : jr nz,OSN29

              call COD28

              inc  hl

              ld   (POZIC),hl 

              jp  OSN00

OSN29 ;JMF - JaMpFlag ++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "J" : jr nz,OSN30

      inc hl  : ld  a,(hl) 

      cp  "M" : jr nz,OSN30

      inc hl  : ld  a,(hl) 

      cp  "F" : jr nz,OSN30

              call COD29

              ld   (POZIC),hl 

              jp   GLV00  ;!!!!!-GLV!!!

OSN30 ;W4O - WIN4OPEN ++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN31

      inc hl  : ld  a,(hl) 

      cp  "4" : jr nz,OSN31

      inc hl  : ld  a,(hl) 

      cp  "O" : jr nz,OSN31

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD30     ;НЕТ
              jp  OSN00

OSN31 ;W4C - WIN4CLOSE +++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "W" : jr nz,OSN32

      inc hl  : ld  a,(hl) 

      cp  "4" : jr nz,OSN32

      inc hl  : ld  a,(hl) 

      cp  "C" : jr nz,OSN32

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD31     ;НЕТ
              jp  OSN00

OSN32 ;FOF - FlagOFF +++++++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "F" : jr nz,OSN33

      inc hl  : ld  a,(hl) 

      cp  "O" : jr nz,OSN33

      inc hl  : ld  a,(hl) 

      cp  "F" : jr nz,OSN33

              call COD32

              inc  hl

              ld   (POZIC),hl 

              jp   OSN00

OSN33 ;VIN - VICTORI !!! ++++++++++++++

      ld hl,(POZIC) : ld  a,(hl) 

      cp  "V" : jr nz,OSN34

      inc hl  : ld  a,(hl) 

      cp  "I" : jr nz,OSN34

      inc hl  : ld  a,(hl) 

      cp  "N" : jr nz,OSN34

              inc hl         ;СРАЗУ ПЕРЕ
              inc hl         ;ВЕЛИ ТК ПА
              ld  (POZIC),hl ;РАМЕТРОВ
              call COD33     ;НЕТ
              jp  OSN00

OSN34 ret 

;г====================================¬

;¦БЛОК ОБРАБОТЧИКОВ УПРАВЛЯЮЩИХ КОДОВ ¦

;L====================================-

;    COD00-COD04 НАХОДЯТСЯ В bib02.C

;-------------- COD ##05 ---------------

;  ПРОЦЕДУРА ВЫРЕЗАНИЯ ФОНА ПОД ОКНОМ

COD05  ld   a,(BANK)

       ld   (RAMTMP),a 

       ld   b,7

       call RAM

       ld   a,(COLW) : ld (SPX),a 

       ld   a,(ROWW) : ld (SPY),a 

       ld   a,(LENW) : ld (SPDL),a 

       ld   a,(HGTW) : ld (SPWS),a 

       ld   a,(WINNMB)

       cp   1

       jr   z,COD051

       cp   2

       jr   z,COD052

       cp   3

       jr   z,COD053

       ld   hl,(WINBF0)

       ld   (SPADR),hl

       call CTSP

       xor  a

       inc  a

       ld   (WINNMB),a

       ld   hl,(SPLST)

       ld   (WINBF1),hl

       ld   a,(RAMTMP)

       ld   b,a

       call RAM

       ret 

COD051 ld   hl,(WINBF1)

       ld   (SPADR),hl

       call CTSP

       ld   a,(WINNMB)

       inc  a 

       ld   (WINNMB),a

       ld   hl,(SPLST)

       ld   (WINBF2),hl

       ld   a,(RAMTMP)

       ld   b,a

       call RAM

       ret 

COD052 ld   hl,(WINBF2)

       ld   (SPADR),hl

       call CTSP

       ld   a,(WINNMB)

       inc  a 

       ld   (WINNMB),a

       ld   hl,(SPLST)

       ld   (WINBF3),hl 

       ld   a,(RAMTMP)

       ld   b,a

       call RAM

       ret 

COD053 ld   hl,(WINBF3)

       ld   (SPADR),hl

       call CTSP

       ld   a,(WINNMB)

       inc  a 

       ld   (WINNMB),a

       ld   a,(RAMTMP)

       ld   b,a

       call RAM

       ret 

RAMTMP defb 0

WINNMB defb 0

WINBF0 defw #C000

WINBF1 defb 0,0

WINBF2 defb 0,0

WINBF3 defb 0,0

;-------------- COD ##06 ---------------

;    ПРОЦЕДУРА ПОЛНОЙ УСТАНОВКИ ОКНА

COD06  call COD01

       call COD03

       call COD04

       ret 

;-------------- COD ##07 ---------------

; ВОССТАНАВЛИВАЕТ СОХРАНЕННЫЙ ФОН ОКНА

COD07  ld   a,(BANK)

       ld   (RAMTMP),a 

       ld   b,7

       call RAM

       ld   a,(COLW) : ld (SPX),a 

       ld   a,(ROWW) : ld (SPY),a 

       ld   a,(LENW) : ld (SPDL),a 

       ld   a,(HGTW) : ld (SPWS),a

       ld   a,#FE    : ld (SPREJ),a 

       ld   a,(WINNMB)

       cp   2

       jr   z,COD072

       cp   3

       jr   z,COD073

       cp   4

       jr   z,COD074

       ld   hl,(WINBF0)

       ld   (SPADR),hl

       call PTSP

       xor  a

       ld   (WINNMB),a

       ld   a,(RAMTMP)

       ld   b,a

       call RAM

       ret 

COD072 ld   hl,(WINBF1)

       ld   (SPADR),hl

       call PTSP

       ld   a,1

       ld   (WINNMB),a

       ld   a,(RAMTMP)

       ld   b,a

       call RAM

       ret 

COD073 ld   hl,(WINBF2)

       ld   (SPADR),hl

       call PTSP

       ld   a,2

       ld   (WINNMB),a

       ld   a,(RAMTMP)

       ld   b,a

       call RAM

       ret 

COD074 ld   hl,(WINBF3)

       ld   (SPADR),hl

       call PTSP

       ld   a,3

       ld   (WINNMB),a

       ld   a,(RAMTMP)

       ld   b,a

       call RAM

       ret 

;-------------- COD ##08 ---------------

;     ПРОЦЕДУРА УСТАНОВКИ ОКНА #1

COD08   ld  a,1

        ld  (COLW),a 

        ld  (ROWW),a 

        ld  a,16

        ld  (LENW),a 

        ld  a,21

        ld  (HGTW),a 

        ld  a,48

        ld  (ATRW),a

        ld  a,2

        ld  (ATX),a : ld (ATX42),a

        ld  (TBP42),a

        ld  a,1

        ld  (ATY),a : ld (ATY42),a 

        xor a 

        ld  (CLR42),a 

        call COD01

        call COD03

        ret 

;-------------- COD ##09 ---------------

;       ПРОЦЕДУРА ВЫВОДА ОКНА #2

;   "ЛИСТОЧКИ" ДЛЯ ВЫБОРА ДЕЙСТВИЯ

COD09 ld a,8 : ld (SPX),a  ;СОХРАНИЛИ

      ld a,19: ld (SPY),a  ;КУСОЧЕК
      ld a,3 : ld (SPDL),a ;ФОНА #1

      ld a,1 : ld (SPWS),a ;27 БАЙТ
      ld hl,38000 : ld (SPADR),hl 

      call CTSP

      ld a,25: ld (SPX),a  ;СОХРАНИЛИ
      ld a,2 : ld (SPY),a  ;КУСОЧЕК
      ld a,2 : ld (SPDL),a ;ФОНА #2

      ld a,2 : ld (SPWS),a ;36 БАЙТ
      ld hl,38030 : ld (SPADR),hl 

      call CTSP

      ld a,12 : ld (ATX42),a ;УСТАНОВИЛИ
                ld (TBP42),a ;ДАННЫЕ ДЛЯ

      ld a,4  : ld (ATY42),a ;ПЕЧАТИ И

      ld a,0  : ld (CLR42),a ;ВСЕ ПАРАМ
      ld a,8  : ld (COLW),a  ;ОКНА И СОХ
      ld a,2  : ld (ROWW),a  ;РАНИЛИ ФОН
      ld a,19 : ld (LENW),a  ;ПОД ВСЕМИ
      ld a,18 : ld (HGTW),a  ;ЛИСТОЧКАМИ
      call COD05             ;

      ld b,3 : call RAM

      ld a,1 : ld (SPREJ),a 

      ld hl,#C000 : ld (SPADR),hl 

      call PTSP

      ld a,8 : ld (SPX),a 

      ld a,19: ld (SPY),a 

      ld a,3 : ld (SPDL),a 

      ld a,1 : ld (SPWS),a 

      ld hl,38000 : ld (SPADR),hl 

      call PTSP

      ld a,25: ld (SPX),a 

      ld a,2 : ld (SPY),a 

      ld a,2 : ld (SPDL),a 

      ld a,2 : ld (SPWS),a 

      ld hl,38030 : ld (SPADR),hl 

      call PTSP

      ld b,0 : call RAM

      ret 

;-------------- COD ##14 ---------------

;  ПРОЦЕДУРА ПЕЧАТИ ТЕКСТА 42 СИМВОЛА

COD14 ld hl,(POZIC)      ;БЕРЕМ ОЧЕРЕД-

      ld a,(hl)          ;НОЙ СИМВОЛ

      cp 27

         jr   nz,CD141   ;#1B-ВЫХОД

         inc  hl         ;=ППС=

         inc  hl 

         ld   (POZIC),hl 

         ret             ;ВЫХОД
CD141 cp 16              ;DLE

         jr   nz,CD142

         ld   a,2

         ld   (RAZR),a 

         inc  hl         ;ПРОП ПРОБЕЛ
         inc  hl         ;НА X

         call CONV

         ld   (ATX42),a 

         ld   (TBP42),a 

         inc  hl         ;НА Y

         call CONV

         ld   (ATY42),a 

         inc  hl 

         ld   (POZIC),hl 

         jr   COD14

CD142 cp 17             ;#11-УСТАНОВКА

         jr   nz,CD143  ;ЦВЕТА

         ld   a,3

         ld   (RAZR),a 

         inc  hl

         call CONV

         ld   (CLR42),a 

         inc  hl 

         ld   (POZIC),hl 

         jr   COD14

CD143 cp 13               ;ПЕРЕВОД
         jr   nz,PR42     ;СТРОКИ
         ld   a,(TBP42)   ;X = TAB

         ld   (ATX42),a   ;

         ld   a,(ATY42)   ;Y = Y+1

         inc  a 

         ld   (ATY42),a 

         inc  hl 

         ld   (POZIC),hl 

         jr   COD14

PR42     push hl 

         ld   l,a

         ld   h,0

         add  hl,hl

         add  hl,hl

         add  hl,hl

         ld   de,(NABOR)

         add  hl,de   ;НАШЛИ СИМВОЛ
         push hl      ;В НАБОРЕ
CHECK    ld   de,(ATX42)

         ld   a,e 

         add  a,a

         ld   e,a

         add  a,a

         add   a,e 

         ld   e,255

         ld   l,8

CHPL1    inc  e

         sub  l

         jr   nc,CHPL1

         add  a,l

         ld   h,a

         ld   a,l

         sub  h

         ld   (SL1-1),a

         ld   (SL2-1),a

ADDRZ    ld   a,d

         and  #18

         add  a,#40

         ld   h,a

         ld   a,d

         and  7

         rrca

         rrca

         rrca 

         add a,e

         ld  l,a

         push hl 

AADDR    ld  a,h

         rrca

         rrca

         rrca

         and #03

         or  #58

         ld  h,a

ATTR     ld   a,(CLR42)   ; ЕСЛИ 0 ТО
         cp   0           ; ОБОШЛИ
         jr   z,ATTR2     ; ИНАЧЕ СТАВИМ
         ld   (hl),a      ; ЦВЕТ
ATTR2    pop  hl 

         pop  de 

         ld   c,8

SCRL     ld   a,(de) 

         ld   b,0

SL1      srl  (hl) 

         djnz SL1

         and  a

         ld   b,0

SL2      rl   a

         rl   (hl)

         djnz SL2

         inc  hl

         ld   (hl),a

         dec  hl

         inc  h

         inc  de 

         dec  c

         jr   nz,SCRL

         pop hl 

         inc hl 

         ld  (POZIC),hl 

         ld  a,(ATX42)

         inc a 

         ld  (ATX42),a 

         jp  COD14

ATX42  defb 0

ATY42  defb 0

CLR42  defb 0

TBP42  defb 0

;-------------- COD ##18 ---------------

;     ПРОЦЕДУРА ВЫРЕЗАНИЯ СПРАЙТА

COD18  inc hl       ;ПЕРЕШЛИ НА ПРОБЕЛ

       inc hl       ;ПЕРЕШЛИ НА 1 ЧИСЛО

       ld   a,5

       ld   (RAZR),a 

       call CONV        ;ПОЛУЧИЛИ АДРЕС
       ld   (SPADR),de  ;КУДА СОХРАНЯТЬ

       inc  hl       ;ПЕРЕШЛИ НА 2 ЧИСЛО

       ld   a,2

       ld   (RAZR),a 

       call CONV

       ld   (SPX),a  ;ПОЛУЧИЛИ X КООРД

       inc  hl 

       call CONV

       ld   (SPY),a  ;Y КООРД
       inc  hl 

       call CONV

       ld   (SPDL),a ;ДЛИНУ
       inc  hl 

       call CONV

       ld   (SPWS),a ;ВЫСОТУ
       push hl 

       call CTSP

       pop  hl 

       ret 

;-------------- COD ##19 ---------------

;       ПРОЦЕДУРА ПЕЧАТИ СПРАЙТА

COD19  inc hl       ;ПЕРЕШЛИ НА ПРОБЕЛ

       inc hl       ;ПЕРЕШЛИ НА 1 ЧИСЛО

       ld   a,5

       ld   (RAZR),a 

       call CONV        ;ПОЛУЧИЛИ АДРЕС
       ld   (SPADR),de  ;КУДА СОХРАНЯТЬ

       inc  hl       ;ПЕРЕШЛИ НА 2 ЧИСЛО

       ld   a,(hl) 

       sub  48

       ld   (SPREJ),a ;РЕЖИМ ЦВЕТА
       inc  hl       ;ПРОП ЗАПЯТУЮ

       inc  hl       ;ПЕРЕШЛИ НА 3 ЧИСЛО

       ld   a,2

       ld   (RAZR),a 

       call CONV

       ld   (SPX),a  ;ПОЛУЧИЛИ X КООРД

       inc  hl 

       call CONV

       ld  (SPY),a   ;Y КООРД
       inc  hl 

       call CONV

       ld   (SPDL),a ;ДЛИНУ
       inc  hl 

       call CONV

       ld   (SPWS),a ;ВЫСОТУ
       push hl 

       call PTSP

       pop  hl 

       ret 

;-------------- COD ##20 ---------------

; ПРОЦЕДУРА ПЕЧАТИ СПРАЙТА В ОСН ОКНО

;        В РАМКУ "ФОТОГРАФИИ"

COD20       ld   a,2

            ld   (RAZR),a

            inc  hl 

            inc  hl

            call CONV

            push hl

            push af 

            ld   b,1

            call RAM

            pop  af 

            ld   b,a 

            ld   hl,#C000

            ld   de,1089

            cp   0

            jr   z,OSN202

OSN201      add  hl,de 

            djnz OSN201

OSN202      ld   (SPADR),hl 

            ld   a,01 : ld (SPREJ),a 

            ld   a,20 : ld (SPX),a 

            ld   a,2  : ld (SPY),a 

            ld   a,11

            ld   (SPDL),a : ld (SPWS),a 

            call PTSP

            ld   b,0

            call RAM

            pop  hl 

            ret 

;-------------- COD ##21 ---------------

; ПРОЦЕДУРА ПЕЧАТИ СПРАЙТА В ОКНО СООБЩ

COD21       ret 

;-------------- COD ##22 ---------------

;ПРОЦЕДУРА ЗАГРУЗКИ ФАЙЛА

COD22 push hl

      ld a,10 : ld (COLW),a  ;СОХРАНИЛИ
      ld a,5  : ld (ROWW),a  ;ФОН  ПОД
      ld a,11 : ld (LENW),a  ;ОКНОМ
      ld a,9  : ld (HGTW),a  ;ЗАГРУЗКИ
      call COD05

      ld a,180: ld (RAMW),a  ;НАРИСОВАЛИ
      ld a,2  : ld (ATRW),a  ;ОКНО
      call COD06

      ld a,(BANK) : ld (RAMTMP),a 

      ld b,3  : call RAM

      ld a,1  : ld (SPREJ),a 

      ld a,12 : ld (SPX),a 

      ld a,6  : ld (SPY),a 

      ld a,7  : ld (SPDL),a 

      ld a,7  : ld (SPWS),a 

      ld hl,#D8E6

      ld (SPADR),hl          ;НАРИСОВАЛИ
      call PTSP              ;ДИСКЕТУ
      ld a,(RAMTMP) : ld b,a 

      call RAM

      pop  hl 

      inc  hl 

      inc  hl  ;ВСТАЛИ НА ИМЯ ФАЙЛА

      push hl 

      ld   bc,#C000

      ld   (LDADR),bc 

      call LOAD

      ld a,10 : ld (COLW),a  ;ВОССТАНОВ
      ld a,5  : ld (ROWW),a  ;ФОН  ПОД
      ld a,11 : ld (LENW),a  ;ОКНОМ
      ld a,9  : ld (HGTW),a  ;ЗАГРУЗКИ
      call COD07

      pop  hl 

      ld   de,9

      add  hl,de 

      ret 

;-------------- COD ##23 ---------------

;  БЕЗУСЛОВНЫЙ ПЕРЕХОД НА ДРУГУЮ СЦЕНУ

COD23  ld   a,2

       ld   (RAZR),a 

       inc  hl 

       inc  hl 

       call CONV

       ld   c,a 

       call FINDSC

       ld   (POZGL),hl 

       ret 

;-------------- COD ##24 ---------------

;          ВЫБОР СЦЕНЫ (1-4)

COD24  ld  a,2

       ld  (RAZR),a 

       xor a 

       ld  (23560),a 

COD241 ld  a,(23560)

       cp  0

       jr  z,COD241

       ld  e,a ;ВРЕМ СОХРАНИЛИ КОД КЛАВИШИ

       ld  a,(hl) 

       sub 48

       ld  d,a ;В d-КОЛИЧЕСТВО ВАРИАНТОВ

       ld  a,e ;ВЕРНУЛИ В a КОД КЛАВИШИ

       cp  "1" :jr nz,COD242

         inc  hl     ;НА ЗАПЯТ

         inc  hl     ;НА ВАР1

         call CONV

         ld   c,a 

         call FINDSC ;НАШЛИ НОВУЮ СЦЕНУ
         ld   (POZGL),hl 

         ret 

COD242 dec d   :jr z,COD241

       cp  "2" :jr nz,COD243

         inc  hl ;НА ЗАПЯТ
         inc  hl ;НА НОМЕР СЦЕНЫ

         inc  hl ;ПЕРВОГО ВАР

         inc  hl ;НА ЗАПЯТ

         inc  hl 

         call CONV

         ld   c,a 

         call FINDSC

         ld   (POZGL),hl 

         ret 

COD243 dec d   :jr z,COD241

       cp  "3" :jr nz,COD244

         inc  hl ;НА ЗАПЯТ
         inc  hl ;НА НОМЕР СЦЕНЫ

         inc  hl ;ПЕРВОГО ВАР

         inc  hl ;НА ЗАПЯТ

         inc  hl ;НА НОМЕР СЦЕНЫ

         inc  hl ;ВТОРОГО ВАР

         inc  hl ;НА ЗАПЯТ

         inc  hl 

         call CONV

         ld   c,a 

         call FINDSC    ; НАШЛИ НОВУЮ СЦЕНУ
         ld   (POZGL),hl 

         ret 

COD244 dec d   :jp z,COD241

       cp  "4" :jp nz,COD241

         inc  hl ;НА ЗАПЯТ
         inc  hl ;НА НОМЕР СЦЕНЫ

         inc  hl ;ПЕРВОГО ВАР

         inc  hl ;НА ЗАПЯТ

         inc  hl ;НА НОМЕР СЦЕНЫ

         inc  hl ;ВТОРОГО ВАР

         inc  hl ;НА ЗАПЯТ

         inc  hl ;НА НОМЕР СЦЕНЫ

         inc  hl ;ТРЕТ ВАР

         inc  hl ;НА ЗАПЯТ

         inc  hl 

         call CONV

         ld   c,a 

         call FINDSC

         ld   (POZGL),hl 

         ret 

;-------------- COD ##25 ---------------

;     ПРОЦЕДУРА ЗАКРЫТИЯ ОКНА #2

;            "ЛИСТОЧКИ"

COD25 ld a,8  : ld (COLW),a 

      ld a,2  : ld (ROWW),a 

      ld a,19 : ld (LENW),a 

      ld a,18 : ld (HGTW),a 

      call COD07

      ret 

;---------------- COD26 ----------------

;ПРОЦЕДУРА ВЫВОДА ОКНА #3 "СКРЕПЫШ"

COD26 ld a,2  : ld (COLW),a  ;СОХРАНИЛИ
                ld (ROWW),a  ;ФОН  ПОД
      ld a,26 : ld (LENW),a  ;ОКНОМ
      ld a,12 : ld (HGTW),a  ;ПОМОЩИ
      call COD05

      ld b,3  : call RAM

      ld a,1  : ld (SPREJ),a 

      ld a,2  : ld (SPX),a 

                ld (SPY),a 

      ld a,7  : ld (SPDL),a 

      ld a,12 : ld (SPWS),a 

      ld hl,#CC07

      ld (SPADR),hl          ;НАРИСОВАЛИ
      call PTSP              ;СКРЕПЫША
      ld a,13 : ld (ATX),a 

      ld a,3  : ld (ATY),a 

      ld a,0  : ld (CLR42),a 

      ld a,9  : ld (COLW),a 

      ld a,2  : ld (ROWW),a

      ld a,19 : ld (LENW),a 

      ld a,12 : ld (HGTW),a 

      ld a,188: ld (RAMW),a  ;НАРИСОВАЛИ
      ld a,96 : ld (ATRW),a  ;ОКНО ДЛЯ
      call COD06             ;ВЫВОДЫ
                             ;ТЕКСТА
      ld b,0  : call RAM

      ret 

;---------------- COD27 ----------------

;ПРОЦЕДУРА ЗАКРЫТИЯ ОКНА #3 "СКРЕПЫШ"

COD27 ld a,2  : ld (COLW),a 

                ld (ROWW),a 

      ld a,26 : ld (LENW),a 

      ld a,12 : ld (HGTW),a 

      call COD07

      ret 

;---------------- COD28 ----------------

;ПРОЦЕДУРА ВКЛЮЧЕНИЯ ФЛАГА

COD28 inc  hl ;ПРОП ПРОБ

      inc  hl ;ВСТАЛИ НА ЧИСЛО

      ld   a,2

      ld   (RAZR),a 

      call CONV

      ld   a,#FF

      ld   bc,FLAG

      ex   de,hl 

      add  hl,bc 

      ld   (hl),a

      ex   de,hl 

      ret 

;-------------- COD ##29 ---------------

;ПЕРЕХОД НА ДРУГУЮ СЦЕНУ,ЕСЛИ ФЛАГ ВКЛЮЧ

COD29  ld   a,2

       ld   (RAZR),a 

       inc  hl  ;ППР

       inc  hl  ;НА ЧИСЛО

       call CONV

       ld   bc,FLAG

       ex   de,hl 

       add  hl,bc 

       ld   a,(hl) 

       ex   de,hl 

       cp   #FF

       jr   z,COD291

          inc  hl 

          call CONV

          ld   c,a 

          call FINDSC

          ld   (POZGL),hl 

          ret 

COD291    inc hl ;ПРОП

          inc hl ;ПЕРВОЕ ЧИСЛО

          inc hl ;ПРОП ЗАПЯТ

          inc hl 

          call CONV

          ld   c,a 

          call FINDSC

          ld   (POZGL),hl 

          ret 

;---------------- COD30 ----------------

;ПРОЦЕДУРА ВЫВОДА ОКНА #4 "ДОСТИЖЕНИЯ"

COD30 ld a,2  : ld (COLW),a  ;СОХРАНИЛИ

                ld (ROWW),a  ;ФОН  ПОД
      ld a,29 : ld (LENW),a  ;ОКНОМ
      ld a,8  : ld (HGTW),a 

      call COD05

      ld b,3  : call RAM

      ld a,1  : ld (SPREJ),a 

      ld a,2  : ld (SPX),a 

                ld (SPY),a 

      ld a,6  : ld (SPDL),a 

      ld a,8  : ld (SPWS),a 

      ld hl,#D735

      ld (SPADR),hl          ;НАРИСОВАЛИ
      call PTSP              ;СПРАЙТ
      ld a,11 : ld (ATX),a 

      ld a,3  : ld (ATY),a 

      ld a,0  : ld (CLR42),a 

      ld a,8  : ld (COLW),a 

      ld a,2  : ld (ROWW),a

      ld a,22 : ld (LENW),a 

      ld a,8  : ld (HGTW),a 

      ld a,196: ld (RAMW),a  ;НАРИСОВАЛИ
      ld a,66 : ld (ATRW),a  ;ОКНО ДЛЯ
      call COD06             ;ВЫВОДЫ
                             ;ТЕКСТА
      ld b,0  : call RAM

      ret 

;---------------- COD31 ----------------

;ПРОЦЕДУРА ЗАКРЫТИЯ ОКНА #4 "ДОСТИЖЕНИЯ"

COD31 ld a,2  : ld (COLW),a 

                ld (ROWW),a 

      ld a,29 : ld (LENW),a 

      ld a,8 : ld (HGTW),a 

      call COD07

      ret 

;---------------- COD32 ----------------

;ПРОЦЕДУРА ВЫКЛЮЧЕНИЯ ФЛАГА
COD32 inc  hl ;ПРОП ПРОБ
      inc  hl ;ВСТАЛИ НА ЧИСЛО

      ld   a,2

      ld   (RAZR),a 

      call CONV

      xor  a 

      ld   bc,FLAG

      ex   de,hl 

      add  hl,bc 

      ld   (hl),a

      ex   de,hl 

      ret 

;---------------- COD33 ----------------

;УНИКАЛЬНАЯ ПРОВЕРКА ТРЕХ УСЛОВИЙ

COD33 ld hl,FLAG+12

      ld a,(hl) 

      cp #FF

      ret nz 

      ld hl,FLAG+18

      ld a,(hl) 

      cp #FF

      ret nz 

      ld hl,FLAG+19

      ld a,(hl) 

      cp #FF

      ret nz 

      ld   c,31

      call FINDSC

      ld   (POZGL),hl

      pop  hl  ;СНИМАЕМ СО СТЕКА АДРЕС

               ;ВОЗВРАТА В OSN33 И

               ;ПРОСТО ПРАГАЕМ НА GLV00

      jp   GLV00

;=======================================

;ПРОЦЕДУРА ПОИСКА СЦЕНЫ

FINDSC  ld   a,c 

        ld   hl,#C000

        cp   0

        ret  z 

FINDSC1 push af 

FINDSC2 inc  hl 

        ld   a,(hl) 

        cp   #FE

        jr   nz,FINDSC2

        pop  af 

        dec  a 

        jr   nz,FINDSC1

        inc  hl   ;НА ИНТЕР

        inc  hl   ;НА ПЕРВ БАЙТ СЦЕНЫ

        ret 

;=======================================

        include "bib02.C"

        include "bib01.C"

;=======================================

POZGL   defb 0,0 ;2БАЙТА!

POZIC   defb 0,0 ;2БАЙТА!

NABOR   defb 0,0 ;2БАЙТА!

COLW    defb 0

ROWW    defb 0

LENW    defb 0

HGTW    defb 0

ATRW    defb 0

RAMW    defb 0

ATX     defb 0

ATY     defb 0

FILEZ1  defb "zast1   C"

FILEZ2  defb "zast2   C"

FILEZ3  defb "zast3   C"

FILEZ4  defb "zast4   C"

FILEOS  defb "osnscr  C"

FILEZ5  defb "gfx00   C"

FILE01  defb "font    C"

FILE03  defb "blsc00  C"

FILE04  defb "blsp00  C"

FILE05  defb "gfx00   C"

